
Change 1 –  fl_SIP_ParseURI
	Function name
	fl_SIP_ParseURI

	Reason for change
	Common function  fl_SIP_ParseURI does not work correctly when parsing tel uri with phone context.

For example:
If px_IMS_CalleeContactUri is set to "tel:+1234;phone-context=test.3gpp.com"

The request line in the  message  expecting this tel SipUrl is incorrectly build via function f_SIP_BuildSipUri_RX as


Record
   scheme := "tel" -- CharacterString
   components Union
      tel Record
         subscriber := "+1234;phone-context=test.3gpp.com" -- CharacterString
   urlParameters := Omit --  Omit
   headers := Omit --  Omit

the correct SipUrl template should be

Record
   scheme := "tel" -- CharacterString
   components Union
      tel Record
         subscriber := "+1234" -- CharacterString
   urlParameters Record
      Record
         id := "phone-context" -- CharacterString
         paramValue Union
            tokenOrHost := "test.3gpp.com" -- CharacterString
   headers := Omit --

	Summary of change
	Function fl_SIP_ParseUri is extended to correctly parse the phone context as urlParameters.

	TTCN module
	IMS_SIP_Templates.ttcn

	MCC160 Comment
	


Before change

  function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var template (present) charstring v_Pattern;

    var template (value) UriComponents v_UriComponents;

    var template (omit) UserInfo v_UserInfo := omit;

    var charstring v_Host;

    var charstring v_Scheme;

    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    v_Pattern  := pattern "([a-z]#(3,4)):(*)";                  // @sic R5s130187 change 8: "pattern" sic@

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern  := pattern "(*)@(*)";                        // @sic R5s130187 change 8: "pattern" sic@

        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern  := pattern "(*):(*)";                      // @sic R5s130187 change 8: "pattern" sic@

          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

        v_UriComponents := cs_SipUriComponent(v_UserInfo, v_Host);

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

        v_UriComponents := cs_TelUriComponent(v_HierPart);

      }
      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_Pattern  := pattern "(*):(*)";                        // @sic R5s130495 Change 11 sic@

        v_UriComponents := cs_UrnUriComponent(regexp(v_HierPart, v_Pattern, 0), regexp(v_HierPart, v_Pattern, 1));

      }

      case else {

        v_UriComponents := cs_OtherComponent(v_HierPart);

      }

    }

    return cs_SipUrl_Common(v_Scheme, v_UriComponents);
  }
After change
  function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var template (present) charstring v_Pattern;

    var template (value) UriComponents v_UriComponents;

    var template (omit) SemicolonParam_List v_ParamList := omit; 
    var template (omit) UserInfo v_UserInfo := omit;

    var charstring v_Host;

    var charstring v_Scheme;

    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    var charstring v_Params;

    var charstring v_Subscriber;

    var charstring v_ParamValue;
    v_Pattern  := pattern "([a-z]#(3,4)):(*)";                  // @sic R5s130187 change 8: "pattern" sic@

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern  := pattern "(*)@(*)";                        // @sic R5s130187 change 8: "pattern" sic@

        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern  := pattern "(*):(*)";                      // @sic R5s130187 change 8: "pattern" sic@

          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

        v_UriComponents := cs_SipUriComponent(v_UserInfo, v_Host);

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

                                                               // Check additional params RFC 4966 cl. 5.4 and 6

        v_Pattern  := pattern "(*);(*)";      // #ob BZ65316             

        v_Params := regexp(v_HierPart, v_Pattern, 1);

        if (v_Params == "") {       

          v_UriComponents := cs_TelUriComponent(v_HierPart);

        } else {

           v_Subscriber := regexp(v_HierPart, v_Pattern, 0);

           v_UriComponents := cs_TelUriComponent(v_Subscriber);

           v_Pattern  := pattern "(*)=(*)";                   

           v_ParamValue := regexp(v_Params, v_Pattern, 1);

           if (v_ParamValue == "") {       

             v_ParamList := {cs_GenericParam(regexp(v_Params, v_Pattern, 0))};

           } else {

             v_ParamList := {cs_GenericParamToken(regexp(v_Params, v_Pattern, 0), v_ParamValue)};

           }

        }

      }  

      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_Pattern  := pattern "(*):(*)";                        // @sic R5s130495 Change 11 sic@

        v_UriComponents := cs_UrnUriComponent(regexp(v_HierPart, v_Pattern, 0), regexp(v_HierPart, v_Pattern, 1));

      }

      case else {

        v_UriComponents := cs_OtherComponent(v_HierPart);

      }

    }

    return cs_SipUrl_Common(v_Scheme, v_UriComponents, v_ParamList); // #ob BZ65316

  }
